Anisotropic compliance and stiffness relations

Write out the x-y two-dimensional compliance matrix and stiffness matrix (Eqn. 3.56) for a single ply of Kevlar/epoxy composite with its fibers aligned 20o from the x axis. Find the stiffness of the ply in the *x* direction

Compliance matrix (Eq. 3.55):

```
> Digits:=4:with(linalg): S:=matrix(3,3,[ [1/E[1], -nu[21]/E[2], 0],
  [-nu[12]/E[1] ,1/E[2],0],[0,0,1/G[12]]]);
```

$$S := \begin{bmatrix} \frac{1}{E_1} & -\frac{\nu_{21}}{E_2} & 0 \\ -\frac{\nu_{12}}{E_1} & \frac{1}{E_2} & 0 \\ 0 & 0 & \frac{1}{G_{12}} \end{bmatrix}$$

Numerical parameters for Kevlar/epoxy:

```
> unprotect(E); E[1]:=80e9; E[2]:=5.5e9; G[12]:=2.1e9; nu[12]:=.31;
  nu[21]:=nu[12]*E[2]/E[1];
```

Compliance matrix evaluated:

```
> S2:=map(eval,S);
```

$$S2 := \begin{bmatrix} .1250\ 10^{-10} & -.3875\ 10^{-11} & 0 \\ -.3875\ 10^{-11} & .1818\ 10^{-9} & 0 \\ 0 & 0 & .4762\ 10^{-9} \end{bmatrix}$$

Transformation matrix (Eq. 3.27):

```
> A:=matrix(3,3,[[c^2,s^2,2*s*c],[s^2,c^2,-2*s*c],[-s*c,s*c,c^2-s^2]
  ]);
```

$$A := \begin{bmatrix} c^2 & s^2 & 2\,s\,c \\ s^2 & c^2 & -2\,s\,c \\ -s\,c & s\,c & c^2 - s^2 \end{bmatrix}$$

Trigonometric relations and angle:

```
> s:=sin(theta);c:=cos(theta);theta:=20*Pi/180;
```

Transformation matrix evaluated:

```
> A2:=evalf(map(eval,A));
```

$$A2 := \begin{bmatrix} .8830 & .1170 & .6430 \\ .1170 & .8830 & -.6430 \\ -.3215 & .3215 & .7660 \end{bmatrix}$$

Reuter's matrix (Eq. 3.30):

```
> R:=matrix(3,3,[[1,0,0],[0,1,0],[0,0,2]]);
```

$$R := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 2 \end{bmatrix}$$

Transformed compliance matrix (Eq. 3.56);

```
> Sbar:=evalf(evalm( R &* inverse(A2) &* inverse(R) &* S2 &* A2 ));
```

$$Sbar := \begin{bmatrix} .6066\ 10^{-10} & -.3220\ 10^{-10} & -.1220\ 10^{-9} \\ -.3223\ 10^{-10} & .1903\ 10^{-9} & .131\ 10^{-10} \\ -.1220\ 10^{-9} & .131\ 10^{-10} & .3629\ 10^{-9} \end{bmatrix}$$

Transformed stiffness matrix (inverse of compliance matrix):

```
> Dbar:=inverse(Sbar);
```

$$Dbar := \begin{bmatrix} .6426\ 10^{11} & .9412\ 10^{10} & .2127\ 10^{11} \\ .9422\ 10^{10} & .6651\ 10^{10} & .2926\ 10^{10} \\ .2127\ 10^{11} & .2923\ 10^{10} & .9795\ 10^{10} \end{bmatrix}$$

Stiffness in *x*-direction is reciprocal of Sbar(1,1):

```
> 'E[x]'=(1/Sbar[1,1])/1e9,' GPa';
```

$$E_x = 16.49,\ GPa$$